The trigger program


� EMBED PowerPoint.Show.4  ���


Comments have been added for the purposes of this document to most of the source code examples below. If more information is required, please do ask. File QCUSTCDT, used as the example database file, is shipped by IBM in library QIWS.


To compile and link the trigger program (assuming an appropriately set current library):


CRTRPGMOD CP01TRG


CRTRPGMOD QCUSTCDT


CRTRPGMOD CP01 (this does no actual processing - it’s just there to satisfy GETPROC’s reference to it)


CRTRPGMOD GETPROC


CRTPGM CP01TRG MODULE(CP01TRG GETPROC CP01 QCUSTCDT)


�
External description of the trigger buffer (trgbuf.pf)


      *Z: CRTPF MBR(*NONE)


                R TRGBUFR


                  TRGFIL        10          COLHDG('File' 'name')


                  TRGLIB        10          COLHDG('Library' 'name')


                  TRGMBR        10          COLHDG('Member' 'name')


                  TRGEVT         1          COLHDG('Trigger' 'event')


                  TRGTIM         1          COLHDG('Trigger' 'time')


                  TRGCMT         1          COLHDG('Commit' 'level')


                  FILL1          3


                  TRGCCS         9B 0       COLHDG('Job' 'CCSID')


                  FILL2          8


                  TRGORO         9B 0       COLHDG('Old' 'record' 'offset')


                  TRGORL         9B 0       COLHDG('Old' 'record' 'length')


                  TRGONO         9B 0       COLHDG('Old' 'null map' 'offset')


                  TRGONL         9B 0       COLHDG('Old' 'null map' 'length')


                  TRGNRO         9B 0       COLHDG('New' 'record' 'offset')


                  TRGNRL         9B 0       COLHDG('New' 'record' 'length')


                  TRGNNO         9B 0       COLHDG('New' 'null map' 'offset')


                  TRGNNL         9B 0       COLHDG('New' 'null map' 'length')


                  TRGVAR         1          COLHDG('Variable part')


�
Main generic trigger program module (cp01trg.rpg)


      * Display file is used for information to display the contents of the


      * trigger buffer on entry to the trigger program. It will obviously be


      * unnecessary in the production trigger program.


     FTRGBUF#   CF   E             WORKSTN


      * TRGDTA is an array overlaying the input parameter TRGBUF (the trigger


      * buffer). TRGDTA@ is a pointer to array TRGDTA. Note that, because TRGDTA is


      * defined as BASED, we will not actually get the 32767 bytes of memory allocated


      * anywhere.


     DTRGBUF         E DS


     DTRGDTA           S              1    DIM(32767) BASED(TRGDTA@)            


     DTRGDTA@          S               *


      * NEWBUF is an array which will, once its basing pointer NEWBUF@ has been set


      * correctly, point to the new record as passed in the trigger buffer.


     DNEWBUF           S              1    DIM(32767) BASED(NEWBUF@)            


     DNEWBUF@          S               *


      * Similarly OLDBUF will eventually contain the old record.


     DOLDBUF           S              1    DIM(32767) BASED(OLDBUF@)            


     DOLDBUF@          S               *


      * The trigger program is passed the length of the trigger buffer as its second


      * parameter. We are not bothered about this value as we are going to rely on


      * external description of the new and old records passed in the trigger buffer.


     DTRGBUFLEN        S              4B 0


      * PROC is a procedure pointer which is set within module GETPROC to point to


      * the special handler procedure for the triggering database file, or to *NULL if


      * there happens to be no handler procedure for this triggering database file.


     DPROC             S               *   PROCPTR


      * Each file has its own module (see the example below for file QCUSTCDT) which


      * takes as input correctly externally described old and new records. From the


      * high-level generic module (the present one) we just pass down what we find in


      * the trigger buffer - we don’t care about its length or content.


     C     PROCPARMS     PLIST


     C                   PARM                    OLDBUF


     C                   PARM                    NEWBUF


      * We have to have a lookup table to translate file name into module name, even


      * though I have set the two the same. This is because you can only CALLB (or


      * point a procedure pointer at) a module whose name is given as a constant or


      * literal, not one whose name is stored in a variable. I have put the lookup


      * table in a module of its own in case at some future time we can think of a


      * better way of handling this.


     C     GETPROCLST    PLIST


     C                   PARM                    TRGFIL


     C                   PARM                    PROC


�
      * Input parameters for the trigger program are the trigger buffer and its


      * length.


     C     *ENTRY        PLIST


     C                   PARM                    TRGBUF


     C                   PARM                    TRGBUFLEN


      * For information in this example, display the contents of the trigger buffer,


      * formatted suitably. Note that the old and new records, and a lot of other


      * information at the end of the trigger buffer, are not in fixed locations in


      * the trigger buffer - we have to use pointers to get at them (see below).


     C                   EXFMT     TRGBUFSCR


      * Get our main TRGDTA array to overlay the passed-in trigger buffer (this


      * probably doesn’t need a pointer, as a simple RPGIV D spec overlay would


      * almost certainly do the job).


     C                   EVAL      TRGDTA@ = %ADDR(TRGBUF)


      * However only a pointer can manage what we do next. Set the old record pointer


      * to point to the correct place in the trigger buffer (using the Old Record


      * Offset field of the trigger buffer).


     C                   EVAL      OLDBUF@ = %ADDR(TRGDTA(TRGORO))


      * And similarly for the new record.


     C                   EVAL      NEWBUF@ = %ADDR(TRGDTA(TRGNRO))


      * Look up the file-specific module name.


     C                   CALLB     'GETPROC'     GETPROCLST


      * If there isn’t one for this file, *NULL is returned and no processing is


      * carried out. Otherwise, pass old and new records to the file-specific module


      * for processing.


     C     PROC          IFNE      *NULL


     C                   CALLB     PROC          PROCPARMS


     C                   END


      * Any other non-file-specific processing could go in here and/or before the call


      * to the file-specific module.


     C                   RETURN


      * Don’t set on LR, for performance.


�
Trigger buffer display file (for information - trgbuf#.dsp)


     A*%%TS  SD  19950719  112540  AMSECOFR    REL-V3R1M0  5763-PW1


     A*%%EC


     A                                      DSPSIZ(24 80 *DS3)


     A          R TRGBUFSCR


     A*%%TS  SD  19950719  112540  AMSECOFR    REL-V3R1M0  5763-PW1


     A                                  1 18'File name:'


     A            TRGFIL    R        O  1 30REFFLD(TRGBUFR/TRGFIL APPLIB/TRGBUF)


     A                                  2 15'Library name:'


     A            TRGLIB    R        O  2 30REFFLD(TRGBUFR/TRGLIB APPLIB/TRGBUF)


     A                                  3 16'Member name:'


     A            TRGMBR    R        O  3 30REFFLD(TRGBUFR/TRGMBR APPLIB/TRGBUF)


     A                                  4 14'Trigger event:'


     A            TRGEVT    R        O  4 30REFFLD(TRGBUFR/TRGEVT APPLIB/TRGBUF)


     A                                  5 15'Trigger time:'


     A            TRGTIM    R        O  5 30REFFLD(TRGBUFR/TRGTIM APPLIB/TRGBUF)


     A                                  6 15'Commit level:'


     A            TRGCMT    R        O  6 30REFFLD(TRGBUFR/TRGCMT APPLIB/TRGBUF)


     A                                  7 18'Job CCSID:'


     A            TRGCCS    R        O  7 30REFFLD(TRGBUFR/TRGCCS APPLIB/TRGBUF)


     A                                  8 10'Old record offset:'


     A            TRGORO    R        O  8 30REFFLD(TRGBUFR/TRGORO APPLIB/TRGBUF)


     A                                  9 10'Old record length:'


     A            TRGORL    R        O  9 30REFFLD(TRGBUFR/TRGORL APPLIB/TRGBUF)


     A                                 10  8'Old null map offset:'


     A            TRGONO    R        O 10 30REFFLD(TRGBUFR/TRGONO APPLIB/TRGBUF)


     A                                 11  8'Old null map length:'


     A            TRGONL    R        O 11 30REFFLD(TRGBUFR/TRGONL APPLIB/TRGBUF)


     A                                 12 10'New record offset:'


     A            TRGNRO    R        O 12 30REFFLD(TRGBUFR/TRGNRO APPLIB/TRGBUF)


     A                                 13 10'New record length:'


     A            TRGNRL    R        O 13 30REFFLD(TRGBUFR/TRGNRL APPLIB/TRGBUF)


     A                                 14  8'New null map offset:'


     A            TRGNNO    R        O 14 30REFFLD(TRGBUFR/TRGNNO APPLIB/TRGBUF)


     A                                 15  8'New null map length:'


     A            TRGNNL    R        O 15 30REFFLD(TRGBUFR/TRGNNL APPLIB/TRGBUF)


     A                                 16 14'Variable part:'


     A            TRGVAR    R        O 16 30REFFLD(TRGBUFR/TRGVAR APPLIB/TRGBUF)


�
Module name lookup (getproc.rpg)


      * First, define constants for all file names we can handle.


     DQCUSTCDT         C                   'QCUSTCDT'


     DCP01             C                   'CP01'


      * Now set up an array (initialised the first time the trigger program is called


      * within a job) of procedure pointers, and another of corresponding file names.


      * We can handle up to 100 file names in this example.


     DPROCDS           DS


     D  PTR1                           *   PROCPTR INZ(%PADDR(QCUSTCDT))


     D  PTR2                           *   PROCPTR INZ(%PADDR(CP01))


     D  PROCS                  1   1600*   PROCPTR DIM(100)


     DFILEDS           DS


     D  FIL1                         10A   INZ(QCUSTCDT)


     D  FIL2                         10A   INZ(CP01)


     D  FILES                  1   1000A   DIM(100)


      * Define the output parameter for the procedure pointer to be passed back.


     DPROC             S               *   PROCPTR


      * Input parameter is file name; output parameter is procedure pointer, or *NULL


      * if there is no file-specific module for this file.


     C     *ENTRY        PLIST


     C                   PARM                    FILE             10


     C                   PARM                    PROC


      * Look up input file name on file names array.


     C                   Z-ADD     1             X                 3 0


     C     FILE          LOOKUP    FILES(X)                               50


      * Return corresponding procedure pointer or *NULL if none.


     C     *IN50         IFEQ      '0'


     C                   MOVE      *NULL         PROC


     C                   ELSE


     C                   MOVE      PROCS(X)      PROC


     C                   END


      * Don’t set on LR, for performance.


     C                   RETURN


Example file-specific module (qcustcdt.rpg)


      * Define old and new records with full external descriptions (note how RPGIV


      * makes differentiating between them dead easy, with the PREFIX keyword).


     DNEW            E DS                  EXTNAME(QCUSTCDT) PREFIX(NEW)


     DOLD            E DS                  EXTNAME(QCUSTCDT) PREFIX(OLD)


      * Old and new records are passed in by reference, but we know what their


      * external descriptions will be.


     C     *ENTRY        PLIST


     C                   PARM                    OLD


     C                   PARM                    NEW


      * So we can work with them at the field level - here is an example.


     C     OLDSTREET     DSPLY


     C     NEWSTREET     DSPLY


      *


     C                   RETURN

